
  Обработка ошибок в Бейсик-программах для ПЭВМ АГАТ-9





Проблема обработки ошибочных и конфликтных ситуаций при написании программ в интерпретаторе Бейсика не случайно выбрана автором в качестве темы данной статьи. Во-первых  это обусловлено ее важностью, так как знания по этому вопросу требуются каждому программисту, желающему создавать законченные программные продукты. Другая сторона проблемы заключается в отсутствии подробной информации по данной теме во всех руководствах по интерпретатору Бейсика ПЭВМ АГАТ-9. В настоящей статье автор постарался собрать по возможности все сведения об организации процесса обработки ошибок на основе отрывочных сведений, почерпнутых из различной литературы и собственных изысканий. Кроме того, статья содержит много полезных советов и конкретных примеров организации обработки ошибок, взятых из успешно работающих программ автора.


Как известно, интерпретатор языка Бейсик ПЭВМ АГАТ-9 имеет встроенные средства проверки ошибок. Подпрограмма проверки анализирует поступающие данные как в режиме непосредственного диалога с программистом, так и при исполнении Бейсик-программ. При возникновении любой ошибочной ситуации эта подпрограмма прерывает нормальную работу интерпретатора и программы пользователя (если она в этот момент исполнялась) и выводит соответствующее ситуации сообщение об ошибке на включенные устройства вывода (в частности на экран монитора). При этом соблюдается следующий формат сообщения об ошибке:





?(TEXT ERROR)		- в случае, если ошибка произошла в диалоге


				  с пользователем


?(TEXT ERROR) В 		- в случае, если ошибка возникла при исполнении


  (LINE NUM)   (LINE TEXT)	  Бейсик-программы





Примечание:


Здесь: 	TEXT ERROR	- сообщение об ошибке


	LINE NUM	- номер строки программы, в которой эта ошибка произошла


	LINE TEXT	- текст строки программы, в которой произошла ошибка


Причем, если работа происходит в алфавитно-цифровом режиме 32*32 символа, то директива или оператор программы, в котором произошла ошибка выделяется красным цветом. В случае если работа происходит в черно-белом режиме дисплея АЦР 64*32, то перед ошибочной директивой или оператором печатается знак “_” (прочерк).





Интерпретатор Бейсика может диагностировать более 20 типов различных ошибок (они приведены в  таблице 1).





Таблица 1. Коды ошибок и их сообщений, диагностируемых интерпретатором Бейсика





Десятичный код ошибки�
Сообщение об ошибке�
�
0�
NEXT БЕЗ FOR�
�
12�
СИНТ. ОШ.�
�
19�
RETURN БЕЗ GOSUB�
�
35�
МАЛО ДАННЫХ�
�
46�
ОШ. ЗНАЧЕНИЕ�
�
57�
ПЕРЕПОЛНЕНИЕ�
�
69�
МАЛО ПАМЯТИ�
�
80�
НЕТ НОМЕРА�
�
90�
ОШ. ИНДЕКС�
�
99�
УЖЕ ОПИСАН�
�
109�
X/0�
�
112�
НЕ В ДИАЛОГЕ�
�
124�
ОШ. ТИП�
�
130�
СТРОКА ДЛИННА�
�
143�
ФОРМУЛА ВЕЛИКА�
�
157�
НАДО RUN�
�
165�
НЕТ МЕТКИ�
�
174�
БАЙТ НЕ ПОЛОН�
�
186�
ОШ. МЕТКА�
�
194�
ОШ. КОД�
�
200�
УЖЕ ЕСТЬ�
�
254�
(ошибка ввода при исполнении команд INPUT, GET)�
�
255�
(прерывание программы нажатием клавиш “УПР-Ц” или “f=“)�
�
�
Появление ошибочной ситуации может быть связано как с непосредственной синтаксической ошибкой в тексте команды (выражения) или с ошибками конструкций, так и с другими причинами (например, возникновением недостатка оперативной памяти). Кроме того, ДОС 3.3 ПЭВМ АГАТ-9, в среде которой работает Бейсик, имеет собственную систему распознавания ошибок, функционально связанную с аналогичной подпрограммой интерпретатора. При вызове функций ДОС из Бейсик-программ возникновение ошибок может быть также обусловлено как синтаксически неправильно построенной командой или сообщенным недопустимым параметром, так и ошибками, возникающими при работе с периферийным оборудованием (например, отсутствием физического устройства, к которому происходит обращение), а также другими ситуациями. Всего ДОС диагностирует 16 типов ошибок (таблица 2).





Таблица 2. Коды ошибок и их сообщений, диагностируемых ДОС 3.3





Десятичный код ошибки�
Сообщение об ошибке�
�
1�
БЕЙСИК НЕ ЗАГРУЖЕН�
�
2,3�
НЕВЕРНОЕ ЗНАЧЕНИЕ�
�
4�
ДИСК ЗАЩИЩЕН�
�
5�
КОНЕЦ ФАЙЛА�
�
6�
ФАЙЛ НЕ НАЙДЕН�
�
7�
ОШ. НОМЕР ДИСКА�
�
8�
ОШИБКА ОБМЕНА�
�
9�
НЕТ МЕСТА НА ДИСКЕ�
�
10�
ФАЙЛ ЗАПЕРТ�
�
11�
ОШ. ДИРЕКТИВА�
�
12�
МНОГО ОТКРЫТЫХ ФАЙЛОВ�
�
13�
ОШ. ТИП ФАЙЛА�
�
14�
НЕТ МЕСТА В ПАМЯТИ�
�
15�
ТОЛЬКО ИЗ ПРОГРАММЫ�
�
16�
НЕТ УСТРОЙСТВА�
�
17�
СИСТЕМА ИСПОРЧЕНА�
�



Для программиста система распознавания и диагностики ошибочных ситуаций является важным средством, позволяющим выявить все ошибки и неточности, допущенные в процессе разработки программы. Однако, когда программа полностью закончена и отлажена программисту следует предпринять еще один важный шаг, необходимый для того, чтобы его творение стало настоящим программным продуктом. Речь идет об организации так называемой подпрограммы обработки ошибок в Бейсик-программе, блокирующей остановку выполнения программы при возникновении конфликтной ситуации (ошибки) системой распознавания и диагностики ошибок среды программирования. При этом тип ошибки может быть распознан собственной подпрограммой обработки ошибок по идентифицирующему коду ошибки, и ее обработка может быть запрограммирована для каждого конкретного места программы отдельно.


Теперь рассмотрим детально те средства организации подпрограмм обработки ошибок, которые предоставляет интерпретатор Бейсика программисту-разработчику. Организация обработки ошибок осуществляется операторами интерпретатора Бейсик ONERR GOTO и RESUME, а также обращением к специальным подпрограммам на машинном языке и ячейкам нулевой страницы оперативной памяти.





1. Оператор ONERR GOTO


Эта команда может использоваться только как оператор и не может быть использована в качестве директивы. Запись команды осуществляется следующим образом:


ONERR GOTO xx,


где xx - номер строки программы, к которой следует перейти при возникновении ошибки.


Оператор ONERR GOTO отключает систему останова программы и запрещает вывод сообщения об ошибке. Этот оператор устанавливает специальный флаг (ячейка $D8=$80 см. таблицу 3), который запрещает перечисленные действия и при возникновении ошибки обуславливает безусловный переход к той строке программы, которая содержится в теле оператора ONERR GOTO. Действие оператора распространяется на всю Бейсик-программу, выполняемую после срабатывания ONERR GOTO. Однако флаг активности подпрограммы обработки ошибок пользователя может быть снят при помощи, например, команды POKE $D8,0. После этого действие оператора ONERR GOTO будет отключено с возвращением к режиму обычного распознавания и диагностики ошибок Бейсика.


�
Если активен флаг оператора ONERR GOTO и в программе имела место ошибка, то после перехода к подпрограмме обработки ошибок, код типа ошибки (Таблицы 1 и 2) будет содержаться в ячейке нулевой страницы оперативной памяти $DE (Таблица 3), а в ячейках $DA-$DB будет содержаться номер строки программы, в которой эта ошибка произошла.


 Примечание: Напомню, что программа пользователя может иметь номера строк в диапазоне 0-65536. Поэтому в указанных ячейках номер строки хранится в формате микропроцессора (соответственно младший/старший байты).


Существуют также и другие, дополнительные указатели на место ошибки в программе, использование которых правда бывает нужно редко (Таблица 3).





Таблица 3.  Использование нулевой страницы памяти при обработке ошибок





Ячейки  памяти�
Значение�
�
$D8�
Флаг команды ONERR GOTO; устанавливается (=$80), если оператор найден в тексте программы, по умолчанию равен 0.�
�
$DA-$DB�
Содержат номер строки программы, в которой произошла ошибка.�
�
$DC-$DD�
Содержат адрес на 1 меньший адреса ячейки ОП, содержащий оператор, при исполнении которого произошла ошибка. Используются оператором RESUME.�
�
$DE�
Код ошибки (см. таблицы 1 и 2)�
�
$DF�
Содержит указатель регистра S микропроцессора (для возвращения из подпрограммы обработки ошибок по RESUME)�
�
$F4-$F8�
Дополнительные указатели для ONERR GOTO�
�
 


2. Оператор RESUME


Команда RESUME также может быть использована как оператор. Специальных параметров она не имеет. 


Этот оператор используется для возвращения из подпрограммы обработки ошибок в основную Бейсик-программу. Он осуществляет возврат к тому самому оператору, в котором произошла ошибка, замеченная оператором ONERR GOTO. При этом предполагается, что подпрограмма обработки ошибок справилась с конфликтной ситуацией и операцию, в которой она произошла, требуется повторить снова.


Следует иметь ввиду, что использование оператора RESUME без предшествующего использования оператора ONERR GOTO может привести к возникновению синтаксической ошибки “СИНТ. ОШ” (код 12), “подвисанию” компьютера или другим странным действиям. Если же ошибка происходит в самой подпрограмме обработки ошибок, оканчивающейся оператором RESUME, то это приведет к зацикливанию этой подпрограммы.





3. Дополнительные сведения об обработке ошибок





3.1 Возникновение ошибок, внутри циклов и подпрограмм


При обработке ошибок, возникших внутри цикла образованного командами FOR ... NEXT, а также в подпрограмме, заключенной между операторами GOSUB ... RETURN, необходимо соблюдать осторожность, так как указатели и стеки адресов возвратов по циклам и подпрограммам в момент ошибки окажутся разрушенными. Поэтому, подпрограмма обработки ошибок после коррекции ошибочной ситуации должна произвести перезапуск цикла (или подпрограммы), для чего она должна осуществить переход к соответствующим операторам FOR (или GOSUB), но ни в коем случае не к операторам NEXT (или RETURN), иначе в последнем случае возникнут ошибки “NEXT БЕЗ FOR” (код 0) или “RETURN БЕЗ GOSUB” (код 19).





3.2 Повышение надежности подпрограммы обработки ошибок


При использовании в Бейсик-программе оператора TRACE, операторов INPUT, GET, PRINT (то есть операторов ввода/вывода) работа подпрограммы обработки ошибок, заканчивающейся оператором RESUME, может стать ненадежной при возникновении N-ного числа последовательных ошибок из-за переполнения аппаратного стека ПЭВМ адресами возврата. Для избежания такой ситуации программисту следует вставить в Бейсик-программу следующую процедуру на языке Ассемблера (Бейсик имеет встроенный миниассемблер) и вызывать ее из подпрограммы обработки ошибок командой CALL:


�
PLA		;Вытолкнуть из стека


TAY		;последний адрес возврата и 


PLA		;сохранить его в A и Y.


LDX $DF	;Загрузить указатель стека по состоянию


TXS		;до возникновения ошибки из ячейки $DF (см. Таблицу 3)


PHA		;Восстановить положение


TYA		;последнего адреса


PHA		;возврата в стеке.


RTS		;Конец подпрограммы.





3.3 Выход из подпрограммы обработки ошибок с помощью оператора GOTO


Подпрограмма обработки ошибок вовсе не обязательно должна заканчиваться оператором RESUME. Устранив конфликтную ситуацию, Вы можете осуществить переход не только к оператору, в котором произошла ошибка, но и к любой другой строке программы. Тем более что в некоторых случаях это просто необходимо (см. пункт 3.1). В этом случае для окончания подпрограммы обработки ошибок следует использовать оператор GOTO xx, где xx- номер строки Бейсик-программы для перехода. Во избежание возможных осложнений перед выходом из подпрограммы обработки ошибок оператором GOTO программисту можно использовать обращение к специальным подпрограммам интерпретатора Бейсика:


$F2AB  -	точка входа в подпрограмму, осуществляющую стирание из внутреннего стека


	управления Бейсика информации, помещенной туда процедурой обработки ошибок


		ONERR GOTO (данных возврата), обнуляет значения всех переменных в программе


$D699  -	точка входа в подпрограмму удаления из внутреннего стека интерпретатора всей


		управляющей информации о возвратах из подпрограммы обработки ошибок, но


		без изменения значений переменных в Бейсик-программе.





3.4 Появление ошибок из-за переполнения оперативной памяти


Отдельной проблемой являются ситуации возникновения ошибок из-за недостатка оперативной памяти. Как эти ситуации могут возникнуть? Обычно они бывают связаны с накоплением большого количества данных в области переменных и расширением ее границ в процессе выполнения программы. Например, это может иметь место при заполнении данными большого массива. Кроме того, по мере выполнения программы, область переменных может расширяться за счет накопления “мусора”, что возможно при изменении старых значений переменных на новые. Ведь в этом случае старые значения переменных из памяти не удаляются; Бейсик просто захватывает новую область памяти под новые значения. Постепенно область переменных может увеличиться настолько, что интерпретатору будет недостаточно ОП для хранения программы, данных и внутренних стеков, вследствие чего возникнет ошибка “МАЛО ПАМЯТИ” (код 69). Особенно быстро это может произойти, если область памяти для переменных принудительно уменьшена операторами LOMEM и HIMEM для каких-то специальных целей.


Бороться с этой проблемой можно несколькими способами. Основными являются использование в подпрограмме обработки ошибок команды CLEAR и функции FRE(x). При этом напомню, что команда CLEAR очищает всю область переменных и сбрасывает все переменные в 0 (для текстовых переменных устанавливается нулевая длина). Также она устанавливает для всех массивов размерность 10 и очищает стек микропроцессора (после ее использования возвращение из подпрограммы обработки ошибок с помощью оператора RESUME невозможно). Кроме того, указанная команда правильно работает только в средах программирования BASIC MASTER (Friends` software, by Serkov S.V. 1994) и BASIC MASTER`95 (Friends` software, by Serkov S.V. 1995). В стандартном интерпретаторе Бейсик А.9.1 (диск ИКП-1) из-за ошибки, допущенной разработчиками, команда CLEAR выполняется неправильно и разрушает имена всех переменных в тексте программы, поэтому при разработке программ под эту среду CLEAR использовать нельзя.


Функция FRE(x) очищает область переменных от накопившихся старых значений переменных освобождая оперативную память, ее применение в подпрограмме обработки ошибок весьма эффективно. Можно использовать эту функцию и просто в тексте основной программы для заблаговременного предупреждения недостатка памяти. Для этого время от времени нужно вставлять в Бейсик-программу выражение вида:





IF (PEEK($70)*256+PEEK($6F))-(PEEK($6E)*256+PEEK($6D))<2 THEN X=FRE(0)


�
Данное выражение условия контролирует размер зоны памяти между окончанием области хранения переменных-массивов и началом области текстовых переменных (то есть собственно свободной области памяти). Когда программа становится настолько большой, что оперативной памяти ей становится недостаточно, указанное выражение производит “чистку мусора”. 





3.5 Особенности обработки ошибок при использовании драйверов Системы расширения возможностей интерпретатора Бейсика через свободное ключевое слово амперсанд (&).


Драйверы СРВИБ (Friends` software 1993-1995), краткое описание которых было приведено в №45 приложения “Информатика” за 1996 год (статья “Система расширения возможностей интерпретатора языка Бейсик ПЭВМ АГАТ-9”, Серков С.В.), в процессе своей работы также могут диагностировать различные ошибки. При этом они спроектированы таким образом, что возникающие при их работе конфликтные ситуации могут быть обработаны в соответствии с основными принципами системы распознавания и обработки ошибок интерпретатора Бейсика. Драйверы СРВИБ могут диагностировать ошибки с кодами и соответствующими сообщениями об ошибках как самого Бейсика (Таблица 1) или ДОС 3.3 (Таблица 2) так и использовать собственные диагностики ошибок. При этом в соответствующих случаях коды ошибок, возникающих при исполнении команд драйверов СРВИБ, как правило, совпадают с аналогичными по значению диагностиками ошибок Бейсика и ДОС. Если имеются какие-либо исключения из этого правила или в драйвере используются сообщения об ошибках с собственными текстами и кодами, то это обязательно отражается в руководстве для программиста по данному драйверу.


При использовании подпрограммы обработки ошибок (оператора ONERR GOTO) коды ошибок драйверов СРВИБ также передаются в ячейку ОП $DE и обработка ошибки происходит по обычным правилам.


Кроме того следует иметь ввиду, что системы программирования BASIC MASTER и      BASIC MASTER`95 имеют встроенный Диспетчер управления драйверами СРВИБ, обеспечивающий их многозадачность. При работе Диспетчера могут также быть диагностированы некоторые сообщения об ошибках и производиться их обработка по правилам обработки ошибок среды "Бейсик" и ДОС. Кроме того, сообщение об ошибке может быть выдано в разной форме в зависимости от того произошла ли ошибка в режиме диалога или при исполнении Бейсик-программы.


Ошибка "УЖЕ ЕСТЬ" (код 200) диагностируется в случае совпадения адреса, переданного на вход Диспетчера при установке нового драйвера СРВИБ в среду (то есть при попытке загрузки нового драйвера на уже занятую область ОП или попытке установить один и тот же драйвер более одного раза).


Ошибка "МАЛО ПАМЯТИ" (код 69) выдается при попытке установки в Диспетчер более 10 драйверов СРВИБ.


Указанные выше сообщения об ошибках выдаются при вызове драйвера СРВИБ с НГМД из Бейсик-программы. В случае аналогичных действий в диалоговом режиме драйвера самостоятельно диагностируют "ОШИБКу УСТАНОВКИ".


Ошибка "ОШ. ЗНАЧЕНИЕ" (код 46) диагностируется при попытке включения командой "DR=" драйвера с номером большим количества установленных в таблицу Диспетчера.





3.6 Организация ошибки пользователя


К сожалению, интерпретатор Бейсика ПЭВМ АГАТ-9 не обладает возможностью организации ошибки пользователя с собственным кодом и сообщением. Доступно лишь организовать ошибку из перечня ошибок самого Бейсика. Сделать это можно, обратившись к точке входа в подпрограмму вызова ошибок интерпретатора (адрес точки входа $1844). При входе регистр X микропроцессора должен содержать код желаемой ошибки (Таблица 1 с учетом перевода значений кодов ошибок в шестнадцатиричную систему счисления). Для реализации этого следует написать подпрограмму на языке Ассемблера вида:





10 *$3000:


20 	! ERROR:LDX $D6


    	!		JMP $1844


	!:





и вызывать ее, когда нужно из Бейсик программы, помещая в ячейку $D6 код желаемой ошибки. Например, так:





550 IF A>6 THEN POKE $D6,46 : CALL ERROR


�
В нашем примере, если переменная A будет иметь значение большее 6, то в строке программы 550 произойдет ошибка “ОШ. ЗНАЧЕНИЕ” (код 46). Ячейка $D6 в примере используется для передачи кода ошибки машинной подпрограмме.


Организовать ошибку ДОС подобным образом невозможно.


Ошибку пользователя с произвольным кодом и сообщением организует процедура &”ERROR” из Библиотеки СРВИБ фирмы Friends` software. Естественно воспользоваться ей могут только программисты, владеющие техникой программирования на языке Ассемблера, так как для ее использования необходимо собрать собственный драйвер СРВИБ. Библиотека СРВИБ содержит и другие полезные команды для организации процесса обработки ошибок, который в самом интерпретаторе Бейсика оставляет желать много лучшего (подробно об этих возможностях можно узнать из специальных описаний Библиотеки СРВИБ).





4. Наиболее часто всречающиеся ошибки


Если в Бейсик-программе используется работа с операционной системой ДОС 3.3 (а это, как правило, имеет место почти в каждой программе), то все типы ошибок ДОС могут встречаться (см. Таблицу 2). Исключение составляют ошибки с кодами 1,2,3,11,15. Ошибки, имеющие коды 7 и 14 встречаются крайне редко. Обработка остальных ошибок ДОС обычно сводится к организации ожидания нажатия клавиши пользователем, чтобы дать ему возможность ликвидировать конфликтную ситуацию. Например, ошибка ДОС ”ОШИБКА ОБМЕНА” может произойти по причине отсутствия ГМД в дисководе. После разрешения конфликта подпрограма обработки ошибок должна вернуться в основную программу для повторения операции ДОС, при исполнении которой произошла ошибка.


Из сообщений об ошибках самого интерпретатора (Таблица 1) наиболее часто встречаются следующие:





МАЛО ПАМЯТИ (69)


	Следует устранить проблему недостатка оперативной памяти используя функцию FRE(X), а 	если потребуется то и оператор CLEAR, с учетом соображений, изложенных в пункте 3.4.


Ошибка ввода в операторах INPUT, GET (254)


	В этом случае лучше очистить буфер клавиатуры оператором POKE $C010,0 и повторить 	ввод, возвратясь в основную программу оператором RESUME


Попытка прерывания исполнения программы нажатием клавиш ”УПР-Ц” или ”f=” (255)


	Следует действовать аналогично предыдущему пункту.





Вообще советов ”на все случаи жизни” дать невозможно. Обработка ошибок в конкретной программе должна строиться программистом с учетом общих правил и главное специфики самой программы





В заключение статьи автор предлагает вниманию читателей пример подпрограммы обработки ошибок. Пример представляет собой шаблон подпрограммы обработки наиболее типичных и часто возникающих конфликтных ситуаций. Шаблон является универсальным и после надлежащей доработки (автор обычно снабжал свои подпрограммы обработки ошибок оконным интерфейсом и звуковым оформлением, используя возможности драйверов СРВИБ) может быть применен в конкретной Бейсик-программе. Пример же всегда является лишь примером и ни в коем случае не претендует на эталон. Автор надеется, что по мере роста опыта программирования и усложнения Ваших программ, уважаемые читатели смогут самостоятельно неограниченно совершенствовать данную подпрограмму или предложат собственные варианты обработки ошибок.





8000	ERKOD=PEEK($DE): ERRLINE=PEEK($DB)*256+PEEK($DA)


8002	REM ERKOD содержит код ошибки


8003	REM ERRLINE - номер строки, в которой она произошла


8005	IF ERKOD<18 THEN ON ERKOD GOTO 8011, 8012,  8013, 8014, 8015, 8016, 8017, 8018, 	8019,8020,8021,8022,8023,8024,8025,8026,8027


8006	IF ERKOD=46 THEN X=FRE(0):IFX<256 THEN CLEAR:RUN


8007	IF ERKOD>=254 THEN POKE $C010,0: RESUME


8008	REM ... обработка других ошибок ...


8011	GOTO 8040 : REM ошибки ДОС с кодами 1, 2, 3 (см. Таблицу 2) не бывает  


8012	GOTO 8040 : по понятным причинам


8013	GOTO 8040�
8014	A$=”ДИСК ЗАЩИЩЕН”: GOTO 8030


8015	A$=”КОНЕЦ ФАЙЛА”: GOTO 80308016	A$=”ФАЙЛ НЕ НАЙДЕН”: GOTO 8030


8017	A$=”ОШ. НОМЕР ДИСКА”: GOTO 8030


8018	A$=”ОШИБКА ОБМЕНА”: GOTO 8030


8019	A$=”НЕТ МЕСТА НА ДИСКЕ”: GOTO 8030


8020	A$=”ФАЙЛ ЗАПЕРТ”: GOTO 8030


8021	A$=”ОШ. ДИРЕКТИВА”: GOTO 8030


8022	A$=”МНОГО ОТКРЫТЫХ ФАЙЛОВ”: GOTO 8030


8023	A$=”ОШ. ТИП ФАЙЛА”: GOTO 8030


8024	A$=”НЕТ МЕСТА В ПАМЯТИ”: GOTO 8030


8025	A$=”ТОЛЬКО ИЗ ПРОГРАММЫ”: GOTO 8030


8026	A$=”НЕТ УСТРОЙСТВА”: GOTO 8030


8027	A$=”СИСТЕМА ИСПОРЧЕНА”


8030	VTAB 30: HTAB (32-LEN(A$))/2: PRINT A$


8031	PRINT: PRINT ”НАЖМИТЕ ЛЮБУЮ КЛАВИШУ”;: WAIT $C000,$80: POKE $C010,0


8040	GOTO 100 : REM переход к строке восстановления рабочего вида программы


	до возникновения ошибки





С пожеланиями удачи в программировании:	Серков С.В.


							Friends` software 

